Q:

Cursor movement too fast for code to read

I have created a MOUSE_MOVE MouseEvent and a code to draw circles while I move the mouse cursor.  The problem is, it doesn't draw out every single circle if I move the mouse too fast.  

Here are the codes I have for the MOUSE_MOVE event. 
stage.addEventListener(MouseEvent. MOUSE_MOVE, mCursor);

public function mCursor(e:MouseEvent):void
{

    var cursor:Shape = new Shape();

    cursor.graphics.beginFill(1, 1);
    cursor.graphics.drawCircle(e.stageX, e.stageY, 10);
    cursor.graphics.endFill();
    addChild(cursor);
}

Would there be an arithmetic equation or physics formula to have it add every single circle such that it can draw a straight line without the blanks in between? 

A:

Just use
cursor.graphics.lineTo(…);

To draw a continuous line between points instead of adding discrete individual circles. 

I erased the above codes and just added this one line of code cursor.graphics.lineTo(e.localX, e.localY); I tested it and there were blanks in between

You have to set the line width first by calling lineStyle() method of the graphics object.  Otherwise the line width is zero (its default value). 
Here's a full working document class:
package 
{
    import flash.display. Sprite;
    import flash.display. Shape;

    import flash.events. MouseEvent;

    public class Main extends Sprite 
    {
        private var cursor:Shape;

        public function Main() 
        {
            cursor = new Shape();
            cursor.graphics.lineStyle(2);
            addChild(cursor);            

            stage.addEventListener(MouseEvent. MOUSE_MOVE, mCursor);
        }

        private function mCursor(e:MouseEvent):void
        {
            cursor.graphics.lineTo(e.stageX, e.stageY);
        }      
    }
}

You likely have to fiddle around with moveTo in order to set an appropriate starting position.  As the code is now, it starts at 0/0.